
Introduction
You probably think that PL/SQL is rather ordinary programming language. Well, it truly is a very powerful and straightforward language, but there are some features "less traveled" which can seem nothing less than weird. This presentation explores some of  stranger nooks and crannies of the PL/SQL language, perhaps in the process making them a little bit less weird to the everyday programmer.

All the code I reference in this paper is available at my PL/SQL Obsession site: www.ToadWorld.com/SF. Just click on the "Trainings" link and then click on the "demo.zip" link.

Disclosure: I will poke fun at Oracle, the PL/SQL language, and implicitly the folks who are responsible for building and enhancing PL/SQL. Please know that it is all done in a spirit of deep appreciation for what PL/SQL (and its developers) has done for me and so many others around the world. But, hey, there's always room for improvement and you've got to keep a sense of humor about all this stuff!

Error Codes: Negative or Positive? Make up your mind, Oracle!

I really would have thought that Oracle would have sorted this out by now. Here's the issue:
When you see this string:

ORA-01855

do you interpret the "-" character as a hyphen or a negative sign? 

I always considered it to be a negative sign, and that just about every single error code in the world of Oracle is negative. The only exceptions are 1 (user-defined exception) and 100 ("No data found" – which is another little weirdness in PL/SQL. The "No data found" exception has two error codes: 100 and -1403).

Yet sometimes Oracle treats the error codes as positive numbers, most notably when using the SAVE EXCEPTIONS clause of FORALL. When you include SAVE EXCEPTIONS, Oracle "saves up" any exceptions it encounters as it executes all the DML statements specified by the bind array. Then if at least one error occurred, it raises the ORA-24381 exception (Argh! Sometimes I worry about all the little bits of information that are stuck in my head!). It also populates the SQL%BULK_EXCEPTIONS pseudo-collection with all the exceptions that were raised. 

Here's an example of a program that uses this feature:

/* bulkexc.sql */

DECLARE
   bulk_errors          EXCEPTION;
   PRAGMA EXCEPTION_INIT (bulk_errors, -24381);
   TYPE namelist_t IS TABLE OF VARCHAR2 (1000);
   enames_with_errors   namelist_t
      := namelist_t ('ABC'
                   , 'DEF'
                   , NULL
                   , 'LITTLE'
                   , RPAD ('BIGBIGGERBIGGEST', 250, 'ABC')
                   , 'SMITHIE'
                    );
BEGIN
   FORALL indx IN enames_with_errors.FIRST .. enames_with_errors.LAST SAVE EXCEPTIONS
      UPDATE employees
         SET last_name = enames_with_errors (indx);
   ROLLBACK;
EXCEPTION
   WHEN bulk_errors
   THEN
      FOR indx IN 1 .. SQL%BULK_EXCEPTIONS.COUNT
      LOOP
         DBMS_OUTPUT.put_line (   'Error '
                               || indx
                               || ' occurred on index '
                               || SQL%BULK_EXCEPTIONS (indx).ERROR_INDEX
                               || ' with error code '
                               || SQL%BULK_EXCEPTIONS (indx).ERROR_CODE
                              );
      END LOOP;
      ROLLBACK;
END;
/
And this is what I see when I run the program:
Error 1 occurred on index 3 with error code 1407

Error 2 occurred on index 5 with error code 12899
Hmmm. Oracle returns the error code as a positive number! So if I want to use SQLERRM to look up the error message for that code, I have to multiply it by -1:

FOR indx IN 1 .. SQL%BULK_EXCEPTIONS.COUNT

LOOP

   DBMS_OUTPUT.put_line

                      (   'Error '

                       || indx

                       || ' occurred on index '

                       || SQL%BULK_EXCEPTIONS (indx).ERROR_INDEX

                       || ' with error '

                       || SQLERRM

                                (  -1

                                 * SQL%BULK_EXCEPTIONS (indx).ERROR_CODE

                                )

                      );

END LOOP;

And then I see this output:

Error 1 occurred on index 3 with error ORA-01407: cannot update () to NULL

Error 2 occurred on index 5 with error ORA-12899: value too large for column  

It would be nice if Larry would decide once and for all if error codes are positive or negative, and then lay down the law.
( 
SQLERRM and DBMS_OUTPUT.PUT_LINE
Speaking of SQLERRM....let's talk about everybody's favorite program: DBMS_OUTPUT.PUT_LINE. For many, many years – all the way up to the release of Oracle Database 10g Release 2, if you tried to display a string of more than 255 characters with DBMS_OUTPUT.PUT_LINE, that built-in would raise an exception. This caused no end of teeth-gnashing and anguished moans from developers over the year. Fortunately, that restriction was lifted in Oracle Database 10g Release 2 – you can now display up to 32K characters. 

But there is still a problem with SQLERRM. We all know about SQLERRM. You call it to return the error message for the current error (obtained by calling SQLCODE). Did you know, however, that Oracle recommends that you not use this function,. and instead call DBMS_UTILITY.FORMAT_ERROR_STACK? 

Why would that be? The problem is that SQLERRM may truncate your error message. In earlier versions of Oracle, truncated occurred at 255 characters. Now, it is 512. DBMS_UTILITY.FORMAT_ERROR_STACK, on the other hand, returns strings of up to 2000 bytes. 
I can just picture some developer years and years ago, confronted with a nasty problem:

He has found that SQLERRM can return error messages that get quite long, depending on application-specific information, like the names of identifiers. And when you try to display that string with DBMS_OUTPUT.PUT_LINE, an exception is raised.
What's a developer to do?

1. Fix DBMS_OUTPUT.PUT_LINE so it displays longer strings.

or

2. Truncate your error message.

Well, obviously, the solution is to truncate the error message!

Isn't that weird?

I am so glad the PL/SQL team finally got around to fixing DBMS_OUTPUT.PUT_LINE! And according to the PL/SQL development manager at the time, all they had to was change one number!
Oracle's built-in string parsing program – No, not really?
Given that PL/SQL is a professional programming language and given that just about every developer at one point or another needs to parse a delimited string, we would fully expect that Oracle would provide a built-in program to do this for us.
And they do – sort of.

It is called DBMS_UTILITY.COMMA_TO_TABLE and here's an example of using it:

DECLARE
   l_list        DBMS_UTILITY.uncl_array;
   l_num_items   PLS_INTEGER;
BEGIN
   DBMS_UTILITY.comma_to_table ('a,b,c,d', l_num_items, l_list);
   DBMS_OUTPUT.put_line (   'Number of elements in list = '
                         || l_num_items
                         || '-'
                         || l_list.COUNT
                        );
   FOR indx IN 1 .. l_list.COUNT
   LOOP
      DBMS_OUTPUT.put_line (l_list (indx));
   END LOOP;
END;
That looks reasonable, doesn't it? Pass it a comma-delimited string and it returns a collection of type DBMS_UTILITY.UNCL_ARRAY and the number of items in the collection.

Well....the problem is, this procedure is just kind of weird. First, check out the output from this program

Number of elements in list = 4-5

a

b

c

d

d

Hmmm. So the procedure passes back a collection with five elements in it (the last element is repeated), but the argument returning the number of elements is correct: 4.

But that's just the start of the weirdness of COMMA_TO_TABLE. Allow me to elaborate:

· It only works with comma-delimited strings. What about carriage return, pipe ("|"), tabs, you name it?
· It will raise an exception if any of the elements in the list is not a valid PL/SQL identifier. I kid you not. 
Check out the following block and the result when I run it.

/* commatotable.tst */

DECLARE
   str     VARCHAR2 (2000);
   tab     DBMS_UTILITY.uncl_array;
   nrows   PLS_INTEGER;
   PROCEDURE parse (thisstr IN VARCHAR2)
   IS
   BEGIN
      DBMS_OUTPUT.put_line ('PARSING "' || thisstr || '" into:');
      DBMS_UTILITY.comma_to_table (thisstr, nrows, tab);
      FOR indx IN 1 .. nrows
      LOOP
         DBMS_OUTPUT.put_line ('   ' || tab (indx));
      END LOOP;
   EXCEPTION
      WHEN OTHERS
      THEN
         DBMS_OUTPUT.put_line ('   ' || DBMS_UTILITY.format_error_stack ());
   END;
BEGIN
   parse ('a,b,c');
   parse ('a123,b#,c$');
   parse ('a123,b456,c^');
   parse ('a,b,c,123');
   parse ('123,456,789');
END;
/
And the results:

PARSING "a,b,c" into:

   a

   b

   c

PARSING "a123,b#,c$" into:

   a123

   b#

   c$

PARSING "a123,b456,c^" into:

   ORA-20001: comma-separated list invalid near ,c^

PARSING "a,b,c,123" into:

   ORA-00931: missing identifier

PARSING "123,456,789" into:

   ORA-00931: missing identifier

Now, that's worse than weird, it's just downright wrong.

I encourage you to visit my www.ILovePLSQLAnd.net site to help me encourage the PL/SQL development team to prioritize adding a real string parsing program to PL/SQL.

In the meantime, the file named "parse.pkg" in my demo.zip archive offers a generic string parsing utility: any delimiter, containing any sorts of characters. Enjoy!
Varray – least varrying of all collection types

I enjoy collections greatly, use them all over my application code. They are efficient and they are flexible (especially string-indexed associative arrays). If you are not already very familiar with collections and applying them every which way, you should prioritize study of collections. 

Unfortunately, collections are not the easiest things on which to  become an expert. There are three different types of collections, each with their own characteristics, strengths and weaknesses. Here is a quick reference:

	associative arrays
	can be used only within PL/SQL blocks
index values can be positive integers, negative integers or strings

can be sparse (index values do not have to be used sequentially)

index values range from -2,147,483,647 to 2,147,483,647

	nested tables
	part of the object model, you must initialize them with constructor functions

must extend the nested table before you assign data to an index value

can be used as the type of a database column (yes, Oracle officially supports denormalization of data

index values range from 1 to 2,147,483,647

	varray or varying array
	part of the object model, you must initialize them with constructor functions

must extend the nested table before you assign data to an index value

can be used as the type of a database column (yes, Oracle officially supports denormalization of data

you must specify the maximum number of elements allowed in the varray at the time of declaration of the type (in Oracle 10g, you can change that size after the initial declaration)


To drive home that last point about varrays, here is a (silly) example of using varrays to implement China's one child policy in an Oracle table:

CREATE OR REPLACE TYPE first_names_t IS VARRAY (2) OF VARCHAR2 (100);
/
CREATE OR REPLACE TYPE child_names_t IS VARRAY (1) OF VARCHAR2 (100);
/
CREATE TABLE family (
   surname VARCHAR2(1000)
 , parent_names first_names_t
 , children_names child_names_t
)
/
DECLARE
   parents    first_names_t := first_names_t ();
   children   child_names_t := child_names_t ();
BEGIN
   parents.EXTEND (2);
   parents (1) := 'Illustrio';
   parents (2) := 'Chantrina';
   children.EXTEND;
   children (1) := 'Gossamer';
   INSERT INTO family
               (surname, parent_names, children_names
               )
        VALUES ('Razmatazz', parents, children
               );
END;
/
So what's weird about this? Very simply this: it seems to me that a varray is the least varying of all the collection types. Wouldn't a better name for this type be the "fixed array" or "limited array"? 
$IF you_wrote_this $THEN who_could_understand_it $END

Some weird stuff in PL/SQL is only weird until you get over the initial....weirdness. And then you see just how brilliantly useful it is. That goes for conditional compilation, whichadds another layer of code to your code: $-prefixed code to be specific.
With conditional compilation, you rite "meta-code that" will be evaluated/processed before your program is compiled. You use $IF to tell the compiler which lines of code you want to be compiled. You use $ERROR to raise a compile time error for whatever reason you choose. You use two $ symbols together to define what are essentially conditional compilation constants that can be referenced in $IF and $ERROR statements.

Sounds strange, doesn't it? Let's look at a few simple examples:

More flexible tracing

The way that most of us trace or debug our code is to drop in calls to DBMS_OUTPUT.PUT_LINE all over the place and hope that something useful pops out. And then we have finished testing, we of course do not want to leave all those trace statements in place – they're kind of embarrassing. So we go back and remove them. Then tomorrow comes, another bug arrives, and we find ourselves going through the same process again. Very unproductive.
With conditional compilation, I can add trace calls to my programs (best not to use DBMS_OUTPUT.PUT_LINE, by the way), but embed them within $IF clauses so that I can decide when and if to include that tracing logic. I can also much more flexibly turn off commits, something we often want to do during testing. And that's what you see below.

ALTER SESSION SET PLSQL_CCFLAGS = 

   'oe_debug:true, oe_trace_arguments:true, commit_OFF:true'
/
CREATE OR REPLACE PROCEDURE calculate_totals (id_in IN PLS_INTEGER)
IS
   l_total PLS_INTEGER;
BEGIN
$IF $$oe_trace_arguments
$THEN
   q$error_manager.trace ('calculate_totals id_in', id_in);
$END            

   lots_of_application_code (l_total);
$IF $$oe_debug 

$THEN
   q$error_manager.trace ('calculate_totals intermediate total', l_total);
$END   

   lots_more_application_code (l_total);
$IF $$commit_off
$THEN
   /* Commit disabled during testing... */
$ELSE
   COMMIT;
$END   

END calculate_totals;
/
Here's another one: I want to stop a particular program from compiling unless the optimization level is set to 2 or higher. Otherwise, the program will run very inefficiently. No problem, $ERROR to the rescue:

CREATE OR REPLACE PROCEDURE compute_intensive_program
IS
BEGIN
$IF $$PLSQL_OPTIMIZE_LEVEL < 2
$THEN
   $ERROR 

      'compute_intensive_program must be compiled with maximum optimization!'
   $END
$END
   lots_of_code_here;
END compute_intensive_program;
/
Finally, conditional compilation can also be used to write one body of code that must run on multiple versions of Oracle, but also automatically takes advantage of the latest features in a given version of Oracle. Just reference the very handy DBMS_VERSION package. For example, Oracle Database 10g Release 2 adds the INDICES OF clause to FORALL, helping us avoid the (prior) requirement that the binding array in a FORALL must be sequentially filled:

PACKAGE BODY otn_demo_insert
IS
   PROCEDURE insert_rows ( rows_in IN otn_demo_aat )
   IS
   BEGIN
$IF DBMS_DB_VERSION.VER_LE_10_1
$THEN
   /* Get rid of any gaps in the collection and then

      use that densely-filled version in the FORALL. */
   DECLARE
      l_dense otn_demo_aat;
      l_index PLS_INTEGER := rows_in.FIRST;
   BEGIN
      WHILE (l_index IS NOT NULL)
      LOOP
         l_dense (l_dense.COUNT + 1) := rows_in (l_index);
         l_index := rows_in.NEXT (l_index);
      END LOOP;
      FORALL indx IN 1 .. l_dense.COUNT
         INSERT INTO otn_demo VALUES l_dense (indx);
   END;
$ELSE
   /* Ah...so much easier in Oracle 10g! */
   FORALL indx IN INDICES OF rows_in
      INSERT INTO otn_demo VALUES rows_in (indx);
$END
   END insert_rows;
END otn_demo_insert;
Conditional compilation is an incredibly handy feature, and you just experienced a very inadequate explanation of that feature. For lots more information, check out Bryn Llewellyn's incredible white paper on this topic on the Oracle Technology Network. Just visit OTN and search for "conditional compilation" and the link should up for you.

Finally, the reason I titled this section "$IF you_wrote_this $THEN who_could_understand_it $END" is that my biggest concern about conditional compilation is the reduced readability of code that uses the $ syntax. It's already plenty hard to figure out what we wrote....I hope that tools vendors like Quest Software will enhance their products to help us "interpret" our code for various CC flag settings.
Reserved word? That's not a reserved word!

Check out the following block of code; do you think it will even compile? And if so, when it executes, what will you see? What will happen?

DECLARE
   PLS_INTEGER     VARCHAR2 (1);
   NO_DATA_FOUND   EXCEPTION;

   SYSDATE         BOOLEAN;
BEGIN
   SELECT dummy
     INTO PLS_INTEGER
     FROM DUAL
    WHERE 1 = 2;
   IF PLS_INTEGER IS NULL
   THEN
      RAISE NO_DATA_FOUND;
   END IF;
EXCEPTION
   WHEN NO_DATA_FOUND
   THEN
      DBMS_OUTPUT.put_line ('No dummy!');
END;
/
The answers are, first of all, it does compile without any errors even though I have declared variables named PLS_INTEGER, NO_DATA_FOUND and SYSDATE!

Second, when I run this code, I get an unhandled exception:

DECLARE

*

ERROR at line 1:

ORA-01403: no data found

ORA-06512: at line 5

Wait a minute! How can that be? NO_DATA_FOUND is the only exception that I specifically handled in the block of code!

Well....not really. Or at least, we need to be very clear about which NO_DATA_FOUND we are talking about.

First of all, the reason I can write odd-looking code like this is that PLS_INTEGER, SYSDATE and NO_DATA_FOUND are not reserved words in the PL/SQL language. Instead, they are simply identifiers declared in the STANDARD package, one of the two default packages of PL/SQL. So if an identifier is not resolved in my own application code, Oracle will try to resolve that identifier in STANDARD and DBMS_STANDARD.
If I declare a variable named NO_DATA_FOUND in a block of code, however, then within that block any unqualified references to NO_DATA_FOUND will be resolved to my variable and not to STANDARD.NO_DATA_FOUND.

Now, there's nothing at all weird about the way PL/SQL deals with these issues. It is, though, very weird-looking code! 
Where's my line number?

TO BE COMPLETED
Where's my program name?

TO BE COMPLETED
Zero length strings and string-indexed collections

What is a zero length string? Here's a block of code that assigns a zero length string to the l_empty variable:

DECLARE

   l_empty VARCHAR2(1) := '';

BEGIN

   NULL;

END;

Here is what the PL/SQL User Guide says about zero-length strings:

"PL/SQL treats any zero-length string like a null."

And, you know, that is almost correct. But I ran into some really strange behavior regarding a zero-length string when I was teaching a class in Melbourne, Australia in January 2008.

As you may know, the index value (position) in a collection cannot be NULL, as you see here:
[image: image1.png]DECLARE
TYPE index_by_string_t IS TABLE OF VARCHAR? (100)
INDEX BY VARCHARZ (100):

1_list  index by string_t
2903
Tolist gNULL

‘A definite no-no

TOAD Script Error

‘ORA-DB502: PL/SQL: numeric or value error: NULL index table key value
ORAD6512: atlne 7.

e ) (iowen ] (o | (o ] (oo





That's reasonable. This isn't:

In other words, Oracle did not complain about me assigning a string to the index value "zero length string"! It sure does not, in this instance, treat such a string the same was a NULL.

Hopefully, Oracle will consider this a bug and fix it in the next release.
%ROWTYPE and %TYPE

I find that the PL/SQL team is generally very careful about the way it adds new syntax to the PL/SQL language, so as to make it as intuitive as possible for us to write our programs.

One great example of this is the cursor variable. A cursor variable is a variable that points to a result set (rows and columns of data). You open a cursor variable for a particular SELECT statement (static or dynamic). For example:

DECLARE
   l_employee   employees%ROWTYPE;
   l_cursor     sys_refcursor;
BEGIN
   OPEN l_cursor FOR SELECT * FROM employees;
   LOOP
      FETCH l_cursor INTO l_employee;
      EXIT WHEN l_cursor%NOTFOUND;
   END LOOP;
   CLOSE l_cursor;
END; 

One of the wonderful things about cursor variables is that the PL/SQL team designed it so that after you write the odd "OPEN FOR" syntax, all the code you write to manipulate the cursor variable is exactly the same as that you would write for an explicit cursor. You can use all the same cursor attributes, the CLOSE statement, the FETCH statement. Very elegantly done.
But I can't avoid the feeling the feeling that they missed the boat just a little bit with %TYPE and %ROWTYPE.

We use both of these declaration attributes to "anchor" the type of a variable or constant to another, already-existing type. %TYPE is used for scalars and %ROWTYPE is used for records, as in:

DECLARE
l_employee   employees%ROWTYPE;
   l_last_name employees.last_name%TYPE;
BEGIN
OK, so here's my question:

Why couldn't we simply use %TYPE for all these anchored declarations, as in:

DECLARE

   CURSOR employees_cur IS SELECT * FROM employees;
   l_employee1   employees_cur%TYPE;
   l_employee2   employees%TYPE;
   l_last_name employees.last_name%TYPE;
Surely, the compiler can figure out from the thing to the left of the % character whether it is a scalar, table or cursor. It will certainly complain if we try to use %ROWTYPE with the wrong sort of thing:

I say: the fewer "moving parts" in a language the better. Less to learn, less to remember, more intuitive. 

Now, I admit that this is a rather minor piece of weirdness, but if I'm going to spend so much of my waking moments with this language, well, I earn the right to whine about the very smallest of issues.

